NOAA Technical Memorandum
NMFS-SEFC-153

Generalized Geographic Mapping System for Computer Graphics



APRIL
1985

U. S. DEPARTMENT OF COMMERCE
National Oceanic and
Atmospheric Administration
National Marine Fisheries Service
Southeast Fisheries Center
Galveston Laboratory
Galveston, Texas 77550

NOAA Technical Memorandum NMFS-SEFC-153

Generalized Geographic Mapping System for Computer Graphics

By

Dennis B. Koi

U. S. DEPARTMENT OF COMMERCE
Malcolm Baldrige, Secretary

National Oceanic and Atmospheric Administration
Anthony J. Calio, Acting Administrator

National Marine Fisheries Service
William G. Gordon, Assistant Administrator for Fisheries

APRIL 1985

Technical Memorandums are used for documentation and timely communication of preliminary interim reports, or special-purpose information, and have not received complete formal review, editorial control, or detailed editing.

TABLE OF CONTENTS

NOTICE

This Technical Memorandum should be cited as follows:

Koi, Dennis B. 1985. Generalized geographic mapping system for computer graphics. NOAA Technical Memorandum NMFS-SEFC-153, 39 p.

It can be obtained by writing to the National Technical Information Service, 5258 Port Royal Road, Springfield, Virginia 22161.

INTRODUCTION

The Generalized Geographic Mapping System (GGMS) is a computer program that draws accurately proportioned maps virtually anywhere in the world, utilizing a variety of land mass data bases. It has evolved over several years from a single, narrow application, to a callable subroutine system capable of handling a variety of applications. The GGMS is written in standard FORTRAN IV and is machine-independent except where noted. The GGMS was written to interact with Tektronix[1/] PLOT-10 software and 4010-series terminals. The only required Tektronix software is the Terminal Control System, Level 1. If the user's capabilities include a Tektronix flatbed plotter, then the Tektronix Utility Routines are also required.

The GGMS is very versatile. It not only generates maps of land masses, but it also makes maximum use of the specified available output "screen" regardless of the selected borders of the desired output. Two mapping projections, rectangular and Mercator, are available, as are numerous different methods of drawing the graticule (latitude and longitude line grid).

The heart of the GGMS is the subroutine MAPP. MAPP is the only GGMS subprogram required to be called before user-defined application routines can be called. Essentially, MAPP computes and draws all borders, draws and labels the graticule, and draws a land mass. As a necessary courtesy to the user, MAPP retains certain values pertaining to the drawn output borders in named common regions.

[1/] Mention of trade names or commercial products herein does not constitute endorsement or recommendation for use.

GENERALIZED GEOGRAPHIC MAPPING SYSTEM

The GGMS consists of four FORTRAN subprograms: MAPP, STREAM, LABEL, and TMERCA. Using the GGMS requires creating a main program which in turn calls all necessary Tektronix subroutines, and the GGMS subroutine MAPP. Section 1 discusses Tektronix considerations, Section 2 the subroutine MAPP, and Sections 3-5 the subroutines STREAM and LABEL, and the function TMERCA, respectively. Appendix A contains a figure which graphically depicts various aspects of the GGMS terminology, Appendix B contains examples of MAPP output, Appendix C contains an explanation of the land mass data base form used by MAPP, Appendix D contains the GGMS source code, and Appendix E contains a sample land mass data base listing.

1 Tektronix Considerations

The user should have available the Tektronix PLOT-10 Terminal Control System (TCS) subroutines, and the Tektronix Utility Routines if output is to a Tektronix flatbed plotter such as the model 4662. When accessing TCS routines, certain steps for initialization of the TCS must take place. Likewise, at the termination of TCS usage, certain "clean-up" steps are necessary. There are also special considerations for initialization and termination of the Utility Routines if used.

1.1 Initialization and Termination

Initialization of the TCS requires calling the TCS routine INITT[2/]. This call must be made before any call to other TCS routines or any of the GGMS routines are made. After all TCS, GGMS, and user application routines (that will draw) have been called, the TCS "clean-up" (termination) routine FINITT should be called.

[2/] Refer to the appropriate Tektronix user manuals for explanations of the TCS and Utility Routines.

For example:

```
C....USER MAIN FORTRAN PROGRAM
     CALL INITT3/
     CALL MAPP
     CALL FINITT
     STOP
     END
```

In the case of directing output to a flatbed plotter, several Tektronix Utility Routines would be required. PLINIT is an initialization routine, PLCOPY selects the flatbed plotter copy mode, PLON logically turns the flatbed plotter on, and PLOFF logically turns the flatbed plotter off.

For example:

```
C....USER MAIN FORTRAN PROGRAM
     CALL INITT
     CALL PLINIT
     CALL PLCOPY
     CALL PLON
     CALL MAPP
     CALL PLOFF
     CALL FINITT
     STOP
     END
```

1.2 Windows - Screen and Virtual

An important relationship in any graphics display is the transformation of user-defined data to a physical location on the graphic display device. The discussion herein centers around the CRT (cathold ray tube) screen

3/Arguments are not given here. See appropriate Tektronix user manuals for TCS or Utility Routine arguments and Section 2 of this text for MAPP subroutine arguments.

since other output devices, e.g. flatbed plotter (when copy mode option is set on), are logical extensions of the screen. On the unenhanced Tektronix screen there are 1,024 addressable screen coordinates in the horizontal or X-axis (designated 0 to 1,023) and 781 addressable screen coordinates in the vertical or Y-axis (designated 0 to 780). When a user defines the data output range, the Virtual Window is defined. This Virtual Window will be transformed by TCS to fit the entire physical screen unless a subsection of that screen is desired in which case the user defines a Screen Window. The Virtual and/or Screen Windows may be defined at any time through the use of the TCS routines TWINDO or SWINDO (for Screen) or DWINDO or VWINDO (for Virtual).

In the GGMS, Chart is defined as the entire drawn output including label areas and the outermost borders. Map is defined as that area where data (e.g., land masses) are drawn. Map is wholly contained within Chart. The user selected Map border coordinates define the Virtual Window for Map. The Virtual Window for Chart is computed internally based on the Map Virtual Window. After computing the Virtual Window for Chart so that the user selected available physical screen is maximized, the Screen Window for Chart and Map are computed and defined.

When MAPP draws the Chart, the Virtual and Screen Windows are set for Chart. When MAPP draws within the Map only, the Virtual and Screen Windows are set for Map. The TCS-clipping algorithm depends on these settings. This algorithm prevents lines from being drawn outside of the desired area. For instance, when a land mass is drawn, the defined Map may include only a sub-set of the total land mass data base available. By setting the Virtual and Screen Windows for Map, only those data within Map will be drawn. When labeling within the Chart title-label area, the Virtual and Screen Windows must be set for Chart.

Both Screen and Virtual Windows for Chart and Map can be user set through the previously mentioned TCS routines after the completion of subroutine MAPP. Note, however, that both must be set together for one or the other. Setting the Virtual Window for Chart and the Screen Window for Map, for instance, will cause unpredictable scaling problems.

The values necessary to set Virtual and Screen Windows for either Chart or Map are stored by MAPP in named common regions readily accessible by the user.

They are:

Chart Screen Window (integer variables)

```
COMMON /SCHART/ MNXSC,MXXSC,MNYSC,MXYSC,LNYSC
```

where

MNXSC = minimum X Screen Window Value for Chart

MXXSC = maximum X Screen Window Value for Chart

MNYSC = minimum Y Screen Window Value for Chart

MXYSC = maximum Y Screen Window Value for Chart

LNYSC = Y Screen Window Value for title-label area divider line

Chart Virtual Window (real variables)

```
COMMON /VCHART/ XMNVC,XMXVC,YMNVC,YMXVC,YLNVC
```

where

XMNVC = minimum X Virtual Window Value for Chart

XMXVC = maximum X Virtual Window Value for Chart

YMNVC = minimum Y Virtual Window Value for Chart

YMXVC = maximum Y Virtual Window Value for Chart

YLNVC = Y Virtual Window Value for title-label area divider line

Map Screen Window (integer variables)

```
COMMON /SMAP/ MNXSM,MXXSM,MNYSM,MXYSM
```

where

MNXSM	=	minimum X Screen Window Value for Map
MXXSM	=	maximum X Screen Window Value for Map
MNYSM	=	minimum Y Screen Window Value for Map
MXYSM	=	maximum Y Screen Window Value for Map

Map Virtual Window (real variables)

```
      COMMON /VMAP/ XMNVM,XMXVM,YMNVM,YMXVM,TYMNVM,TYMXVM
```

where

XMNVM	=	minimum X Virtual Window Value for Map
XMXVM	=	maximum X Virtual Window Value for Map
YMNVM	=	minimum Y Virtual Window Value for Map
YMXVM	=	maximum Y Virtual Window Value for Map
TYMNVM	=	projected minimum Y Virtual Window Value for Map
TYMXVM	=	projected maximum Y Virtual Window Value for Map

If no title-label area is drawn (see Section 2.5) then LNYSC = MXYSC and YLNVC = YMXVC.

In the common region VMAP, TYMNVM and TYMXVM will differ from YMNVM and YMXVM, respectively, only if the Mercator projection has been specified (XMNVM and XMXVM, the east and west coordinate boundaries, respectively, are not affected).

At the completion of MAPP (whether or not a land mass was drawn) the Screen and Virtual Windows are set for Map.

2 Subroutine MAPP

Subroutine MAPP draws Chart and Map as specified by the user in the argument list. The calling sequence for MAPP is:

```
    CALL MAPP (SOUTH,NORTH,EAST,WEST,IPROJ,ZINCR,LAND,LAREA,ISX,ISY)
```

where

SOUTH = southern latitudinal border for Map (real argument)
NORTH = northern latitudinal border for Map (real argument)
EAST = eastern longitudinal border for Map (real argument)
WEST = western longitudinal border for Map (real argument)
IPROJ = projection desired (integer argument)
ZINCR = type of graticule drawn and labeling desired (real argument)
LAND = land mass desired (integer argument)
LAREA = title-label area desired (integer argument)
ISX = maximum available physical screen for X (integer argument)
ISY = maximum available physical screen for Y (integer argument)

2.1 SOUTH, NORTH, EAST, WEST

The arguments for SOUTH, NORTH, EAST, and WEST are real and any value can be input. However, in order to output a correctly labeled Map, the interrelations of these values and those for ZINCR (see Section 2.3 on ZINCR for a detailed explanation) must be considered. SOUTH is always a value less than NORTH, and WEST is always a value less than EAST. The precisions of the arguments should not be greater than tenths of a degree.

MAPP operates on the cartesian coordinate system where the origin is the intersection of a parallel and meridian. The Gulf of Mexico land mass data base (see Appendix E) is based on the parallel being the Equator and the meridian being the Prime Meridian (passing through Greenwich). A Map whose outer borders are entirely north of the Equator and east of the Prime Meridian would have all positive arguments for SOUTH, NORTH, EAST, and WEST. A Map whose outer borders are entirely south of the Equator and West of the Prime Meridian would have all negative arguments. The coordinate system for Map can cross the Equator or Prime Meridian or both. The values of NORTH or SOUTH should never exceed +90 or be less than -90. In actuality, +90 and -90 can be approached, but never reached, in a Mercator projection. The values of EAST and WEST should never exceed +180 or be less than -180. Note that MAPP is not capable of interpreting and outputting correct results if the intent is to create a Map that will include a continuous area across the 180° Meridian or across polar zones. A Map could be created to include areas on both sides of the 180° Meridian if the data coordinates were based on a different Prime Meridian other than the Greenwich Prime Meridian. Note, however, that labeling of the graticule will be in the user-designed coordinate system and not based on the Greenwich Prime Meridian.

2.2. IPROJ

The two types of projections available with MAPP are rectangular and Mercator. The GGMS will retain the specified projection after completion of MAPP for user-defined application routines. The options are:

IPROJ = 1	For rectangular projection
IPROJ = any other integer	For Mercator projection

The Mercator is a conformal, cylindrical map projection. Within small area plottings it retains the property of shape. The rectangular is a cylindrical map projection that is similar to the Mercator but has uniform spacing of the parallels. It can be useful in displaying data where distortion of shape is not important.

2.3 ZINCR

The MAPP subroutine offers a variety of options for drawing and labeling the graticule. The options are:

ZINCR = 0.	No lines are drawn. No labeling.
$.1 \leq$ ZINCR $\leq .9$	Whole degree lines are drawn and labeled. Fractions of a degree are tic-marked according

	to the value selected (i.e., if ZINCR = .1, tenths of a degree are tic-marked).
$1. \le$ ZINCR $\le 99.$	Degree lines are drawn and labeled according to the value selected (i.e., if ZINCR = 10., every 10-degree line is drawn and labeled).
$901. \le$ ZINCR $< 999.$	Degree lines are labeled only according to the value selected minus 900. (i.e., if ZINCR = 920., every 20-degree line is labeled).
ZINCR = 999.	Only the borders of the Map are drawn and labeled. If borders with fractional values are specified by the user, labeling is unpredictable.

In all cases except where ZINCR = 0., the NORTH, SOUTH, EAST, and WEST borders will be drawn. They will also be labeled if they fall on a whole degree (except when ZINCR = 999., in which case labeling is forced). In choosing a value for ZINCR, the user should be aware of how evenly (i.e., without a remainder) the value of ZINCR will divide the differences between the NORTH-SOUTH borders and the EAST-WEST borders. Only an even division will cause a correct output Map. Values of ZINCR other than those above will default to ZINCR = 0. The precision of ZINCR should not be greater than tenths.

2.4 LAND

This argument specifies whether or not MAPP will plot the land mass data base attached as logical unit 1. The options are:

LAND = 1	Land mass be be drawn
LAND = any other integer	No land mass will be drawn

Note that an argument of 1 for LAND simply specifies to MAPP to call the subroutine STREAM. For more information on STREAM and drawing land masses see section 3.

2.5 LAREA

This argument specifies the size of a title-label area for the Chart.

The options are:

LAREA = 0	No title-label area drawn
LAREA = 1, 2, ...	A title-label area drawn at the top of the Chart equivalent in size to the argument times the distance between the Map border and the Chart border
LAREA = -1, -2, ..	Same as LAREA = 1, 2, .. except drawn at the bottom of the Chart.

Absolute values of LAREA much greater than 2 cause the title-label area to become disproportionately large compared to the Map itself.

2.6 ISX, ISY

The arguments ISX and ISY are used to define the maximum available output display space in absolute addressable hardware units. Normally, for maximum utilization, these values should be 1023 and 780 respectively. The minimum available display space will always be the origin (0,0). ISX and ISY are used when it desired to draw a physically reduced plot. For instance, if drawing to an 8½" by 11" sheet on the 4662 flatbed plotter, set ISX = 586 and ISY = 780. This will leave a margin of about ½" all around the Chart border to the paper's edge. Chart and Map windows (both Screen and Virtual) are adjusted accordingly.

3 Subroutine STREAM

Subroutine STREAM reads a coded land mass data base of latitude-longitude coordinates and draws them. STREAM first reads a value that is the number of coordinate values to follow in the "stream". The first pair of coordinate values read by STREAM is always a "move-to" point. Succeeding pairs are then "drawn-to". STREAM has no parameters and can be tailored to read land mass data in whatever form desired. See Appendix C for a detailed explanation of the coded data form read by STREAM.

4 Subroutine LABEL

Subroutine LABEL places labels for the graticule outside of the Map border but within the Chart border. This routine is called as needed by MAPP and is generally not called by the user. Labeling varies according to the value selected for ZINCR. The arguments to the subroutine are:

X = Longitude location for the label in Map Virtual Window units
Y = Latitude location for the label in Map Virtual Window units
V = ¼ of amount of space between Map border and Chart border for locating label
IFLG = 1 if label to be on WEST side of Map.
IFLG = 2 if label to be on EAST side of Map.
IFLG = 3 if label to be on SOUTH side of Map.
IFLG = 4 if label to be on NORTH side of Map.

For plots drawn on the Tektronix terminal screen "hardware" characters are used. If the Tektronix Utility Routines are available, enabling "software" characters through the subroutine SWCHAR will generate more pleasing labels on the screen.

5 Function TMERCA

TMERCA is a FORTRAN function and is used in assignment statements. The procedure to invoke TMERCA is:

```
B = TMERCA (A)
```

where

A = The unprojected latitudinal variable.

B = The projected latitudinal variable.

If the chosen projection is rectangular (i.e., if IPROJ = 1), TMERCA has no effect and B=A. If the chosen projection is Mercator, then the value of A will be computed to reflect that projection. TMERCA can, and should, be used in user-defined applications routines. All longitudinal values either "moved-to" or "drawn-to" must be projected by TMERCA when the Map is drawn with a Mercator projection. For example, if the user has drawn a Map and is now point-plotting locations on that Map, the TCS statement for doing this would be:

```
CALL POINTA (X, TMERCA (Y))
```

where X is the longitudinal coordinate and Y the latitudinal coordinate.

PROGRAM NOTES

Every attempt has been made to make the GGMS program code as machine-independent as possible. The connection between the internal logical unit 1 and the external file containing the land mass data are left to the user as this connection varies dependent on the hardware used. Note that there is little error checking of the arguments to Subroutine MAPP. To allow the user the options of easy modification and flexibility, extensive error checking was deemed unnecessary and cumbersome. The author welcomes any suggestions for improvement and/or modifications, and would be interested in learning about applications and innovative uses of this system. Proper acknowledgment of use of this program and/or land mass data base in publications of graphics generated therefrom would be appreciated.

ACKNOWLEDGMENTS

The author is grateful to Frank Patella, Jr. for the original mapping program idea. Also appreciated are the efforts of Julie Jamieson who spent countless hours digitizing the Gulf of Mexico coastline data base used for figures B.1-B.8 and listed in Appendix E. Steven McCommas digitized the Atlantic coastline data base used for figures B.9 and B.10. The TMERCA function was adapted from the MERCTR subroutine provided by the National Oceanographic Data Center, Washington, D.C. Dr. Charles Caillouet and Lorretta Sullivan reviewed the manuscript and provided many helpful suggestions. Daniel Patlan assisted in preparing the cover and Beatrice Richardson typed the manuscript.

APPENDIX A

Terminology

In reference to Figure A (this Appendix):

1. Chart is the entire drawn figure (see Section 1.2 of the text).

2. Map is the area contained in the inner rectangle (see Section 1.2 of the text).

3. All variable names in the common regions are printed along the borders they represent (see Section 1.2 of the text).

4. The possible signs for SOUTH, NORTH, EAST, and WEST are dependent on the quadrant they lie in, and are given in those quadrants defined by the lines labeled "equator" and "prime meridian" depicted within the Map (see Section 2.1 of the text).

5. The title-label area is shown as if a positive value had been chosen for LAREA. Negative values would place the title-label area at the bottom of the Chart (see Section 2.5 of the text).


MXYSC
YMXVC
TITLE-LABEL AREA
LNYSC
YLNVC
MXYSM
(T)YMXVM
NORTH
MNXSC
MNXSM
PRIME MERIDIAN
MXXSM
MXXSC
NORTH=+
SOUTH=+
EAST=-
WEST=-
NORTH=+
SOUTH=+
EAST=+
WEST=+
WEST
EQUATOR
EQUATOR
EAST
PRIME MERIDIAN
NORTH=-
SOUTH=-
EAST=-
WEST=-
NORTH=-
SOUTH=-
EAST=+
WEST=+
XMNVC
XMNVM
XMXVM
XMXVC
MNYSM
SOUTH
(T)YMNVM
MNYSC
YMNVC

Figure A

APPENDIX B

MAPP subroutine usage examples

The figures in this appendix were drawn with argument values for MAPP as listed below for each figure. Figure B.1 shows the entire Gulf of Mexico land mass data base listed in Appendix F.

Figure	SOUTH	NORTH	EAST	WEST	IPROJ	ZINCR	LAND	LAREA	ISX	ISY
B.1	18.0	31.0	-80.0	-98.0	0	999.0	1	0	780	586
B.2	26.0	30.0	-94.0	-98.0	0	1.0	1	0	586	780
B.3	28.9	29.9	-94.4	-95.5	0	0.1	1	0	780	586
B.4	28.0	31.0	-88.0	-94.0	0	901.0	1	0	780	586
B.5	24.0	30.0	-80.0	-84.0	0	2.0	1	0	586	780
B.6	24.0	26.0	-80.0	-84.0	0	0.5	1	0	780	586
B.7	24.0	26.0	-80.0	-84.0	0	0.5	1	2	780	586
B.8	24.0	26.0	-80.0	-84.0	0	0.5	1	-3	780	586
B.9	-50.0	70.0	50.0	-110.0	0	10.0	0	0	780	586
B.10	-50.0	70.0	50.0	-110.0	1	10.0	0	0	780	586

All figures in this appendix are reproduced exactly as drawn.
The cover was drawn with the following argument values for MAPP:

26.0,30.0,-94.0,-98.0,0,1.0,1,8,586,780



Figure B.1



Figure B.2



Figure B.3



Figure B.4



Figure B.5



Figure B.6



Figure B.7



Figure B.8


110 100 90 80 70 60 50 40 30 20 10 0 10 20 30 40 50
70 60 50 40 30 20 10 0 10 20 30 40 50
70 60 50 40 30 20 10 0 10 20 30 40 50
110 100 90 80 70 60 50 40 30 20 10 0 10 20 30 40 50

Figure B.9


110 100 90 80 70 60 50 40 30 20 10 0 10 20 30 40 50
70 60 50 40 30 20 10 0 10 20 30 40 50
110 100 90 80 70 60 50 40 30 20 10 0 10 20 30 40 50
70 60 50 40 30 20 10 0 10 20 30 40 50

Figure B.10

APPENDIX C

Land mass data base form for STREAM

STREAM expects the land mass data to be in the form of blocks or "streams". Each block begins with an integer value representing the number of coordinate values to follow. The first two are not coded. However, each succeeding value is coded in a way that is offset from its preceeding value and is decoded via the following algorithm:

$$x = r + \frac{P-50}{100}$$

where x = the unknown coordinate value

r = the preceeding coordinate value

p = the coded coordinate value

The coded values occur in 4-digit sets, where the first 2 digits of the set represent the latitude and the last 2 digits the longitude.

Given the following stream of data:

12 20.60 -86.76 46383434385462627062

The number of coordinate values is 12.

The first, or starting pair, is 20.60° latitude and -86.76° longitude. Note the minus sign which indicates the western hemisphere. Also note that coordinates are given as decimal fractions and not minutes and seconds.

The second pair of coordinates would decode from the 4-digit set of 4638 as follows:

$$\text{latitude} = 20.60 + \frac{46-50}{100} = 20.56^\circ$$

$$\text{longitude} = -86.76 + \frac{38-50}{100} = -86.64^\circ$$

The third pair of coordinates would decode from the next 4-digit set of 3434 as follows:

$$\text{latitude} = 20.56 + \frac{34-50}{100} = 20.40^{\circ}$$

$$\text{longitude} = -86.64 + \frac{34-50}{100} = -86.80^{\circ}$$

and so on.

Remember that the first coordinate pair in a block (20.60 and -86.64 in this example) is always "moved-to" by STREAM. Each succeeding pair is then "drawn-to".

STREAM make no attempt to seek-out valid (i.e., within the Virtual Map Window) data blocks. It simply tries to draw them all. Only those within the virtual map window are drawn. For small land mass data bases, this is not a problem. However, when the data base is large and only a very small portion of it is being drawn, this method would be inefficient. More complex data management schemes would then be a necessity. Those are left to the user.

Although the coding scheme results in slower execution speed, in many cases considerable disk storage space can be saved. Note that the 2 digit coding scheme for each value restricts the maximum distance between digitized points to 0.49 degrees.

Appendix D

GGMS Source Code Listing

```
C***********************************************************************
C*                                                                     *
C*                            G G M S                                  *
C*                                                                     *
C*               GENERALIZED GEOGRAPHIC MAPPING SYSTEM                 *
C*               =========== ========== ======= ======                 *
C*                                                                     *
C*                      AUTHOR:   DENNIS B. KOI                        *
C*                                                                     *
C***********************************************************************
      SUBROUTINE MAPP(SOUTH,NORTH,EAST,WEST,IPROJ,ZR,LAND,LAREA,ISX,ISY)
C...MAPP COMPUTES AND DRAWS CHART AND MAP FROM USER SPECIFIED
C...COORDINATES.
      LOGICAL TENTH,LINYES
      REAL NORTH
      COMMON /SCHART/ MNXSC,MXXSC,MNYSC,MXYSC,LNYSC
      COMMON /VCHART/ XMNVC,XMXVC,YMNVC,YMXVC,YLNVC
      COMMON /SMAP/   MNXSM,MXXSM,MNYSM,MXYSM
      COMMON /VMAP/   XMNVM,XMXVM,YMNVM,YMXVM,TYMNVM,TYMXVM
      COMMON /PRJECT/ IPROJJ
C...TEKTRONIX SCREEN SIZE IN SCREEN UNITS
C...ISX=MAXIMUM SCREEN UNITS IN X-AXIS DIRECTION
C...CHANGE 1023 BELOW AS NECESSARY FOR LARGER TEKTRONIX PLOTTERS
      IF (ISX.GT.1023) RETURN
C...ISY=MAXIMUM SCREEN UNITS IN Y-AXIS DIRECTION
C...CHANGE  780 BELOW AS NECESSARY FOR LARGER TEKTRONIX PLOTTERS
      IF (ISY.GT.780) RETURN
      IPROJJ=IPROJ
      ZINCR=ZR
C...CHECK SOUTH,NORTH,EAST,WEST
      IF (SOUTH.GT.NORTH.OR.WEST.GT.EAST) RETURN
C...ABOVE CHECK MAY NEED TO BE MODIFIED OR ELIMINATED IF
C...USER HAS DEFINED HIS OWN COORDINATE SYSTEM
      XMNVM=WEST
      XMXVM=EAST
      YMNVM=SOUTH
      YMXVM=NORTH
C...PROJECT SOUTH AND NORTH COORDINATES
      TYMNVM=TMERCA(YMNVM)
      TYMXVM=TMERCA(YMXVM)
C...COMPUTE CHART AND MAP VIRTUAL DIMENSIONS
      T1=YMXVM-YMNVM
      T2=XMXVM-XMNVM
C...CHECK VALIDITY OF ZINCR
      IF (ZINCR.LT.0.) ZINCR=0.
      IF (ZINCR.GE.99.AND.ZINCR.LE.900.) ZINCR=0.
      IF (ZINCR.GT.999.) ZINCR=0.
      IF (ZINCR.GE.1.) GO TO 6
      DEGS1=FLOAT(INT((T1+.01)*10.))/10.
      DEGS2=FLOAT(INT((T2+.01)*10.))/10.
```

```
      GO TO 7
    6 DEGS1=T1+1.0
      DEGS2=T2+1.0
    7 V1=(TYMXVM-TYMNVM)*0.07
      V2=T2*0.07
      VX=(V1+V2)/2.0
      XMNVC=XMNVM-VX
      XMXVC=XMXVM+VX
      IF (LAREA) 8,9,10
    8 YMNVC=TYMNVM-VX*(IABS(LAREA)+1)
      YMXVC=TYMXVM+VX
      YLNVC=TYMNVM-VX
      GO TO 11
    9 YMNVC=TYMNVM-VX
      YMXVC=TYMXVM+VX
      YLNVC=YMXVC
      GO TO 11
   10 YMNVC=TYMNVM-VX
      YMXVC=TYMXVM+VX*(LAREA+1)
      YLNVC=TYMXVM+VX
C...SET CHART VIRTUAL WINDOW
   11 CALL DWINDO(XMNVC,XMXVC,YMNVC,YMXVC)
C...COMPUTE CHART SCREEN DIMENSIONS
      YDIST=YMXVC-YMNVC
      XDIST=XMXVC-XMNVC
      XRATIO=ISX/XDIST
      YRATIO=ISY/YDIST
      IF (XRATIO.LT.YRATIO) GO TO 30
      USEFUL=YRATIO/XRATIO
      USELES=1.0-USEFUL
      IADJ=INT(USELES*ISX/2.0)
      MNXSC=0+IADJ
      MXXSC=ISX-IADJ
      MNYSC=0
      MXYSC=ISY
C...SET CHART SCREEN WINDOW
      CALL TWINDO(MNXSC,MXXSC,MNYSC,MXYSC)
      GO TO 31
   30 USEFUL=XRATIO/YRATIO
      USELES=1.0-USEFUL
      IADJ=INT(USELES*ISY/2.0)
      MNYSC=0+IADJ
      MXYSC=ISY-IADJ
      MNXSC=0
      MXXSC=ISX
C..SET CHART SCREEN WINDOW
      CALL TWINDO(MNXSC,MXXSC,MNYSC,MXYSC)
C..DETERMINE VX IN SCREEN UNITS
C..COMPUTE MAP IN SCREEN UNITS
   31 KUNITS=MXXSC-MNXSC
      ADJ=VX*KUNITS/XDIST
      MNXSM=MNXSC+ADJ
      MXXSM=MXXSC-ADJ
      IF (LAREA) 32,33,34
   32 MNYSM=MNYSC+(IABS(LAREA)+1)*ADJ
```

```
      MXYSM=MXYSC-ADJ
      LNYSC=MNYSC+(IABS(LAREA))*ADJ
      GO TO 35
   33 MNYSM=MNYSC+ADJ
      MXYSM=MXYSC-ADJ
      LNYSC=MXYSC
      GO TO 35
   34 MNYSM=MNYSC+ADJ
      MXYSM=MXYSC-(LAREA+1)*ADJ
      LNYSC=MXYSC-LAREA*ADJ
C...DRAW CHART OUTLINE
   35 CALL MOVEA(XMNVC,YMNVC)
      CALL DRAWA(XMXVC,YMNVC)
      CALL DRAWA(XMXVC,YMXVC)
      CALL DRAWA(XMNVC,YMXVC)
      CALL DRAWA(XMNVC,YMNVC)
      IF (LAREA.EQ.0) GO TO 60
      CALL MOVEA(XMNVC,YLNVC)
      CALL DRAWA(XMXVC,YLNVC)
   60 IF (ZINCR.EQ.0.) GO TO 125
      Z=0.0
      VXX=VX/4.0
      CALL CHRSIZ(3)
      LINYES=.FALSE.
      IF (ZINCR.EQ.999.) GO TO 85
      IF (ZINCR.LT.1.) GO TO 90
C...ZINCR .GE. 1, THEREFORE WHOLE DEGREES DRAWN
      LINYES=.TRUE.
      IF (ZINCR.LT.901.) GO TO 65
      ZINCR=MOD(INT(ZINCR),100)
      LINYES=.FALSE.
   65 CALL LABEL(XMNVM+Z,TYMNVM,VXX,1)
      IF (LINYES) CALL MOVEA(XMNVM+Z,TYMNVM)
      IF (LINYES) CALL DRAWA(XMNVM+Z,TYMXVM)
      CALL LABEL(XMNVM+Z,TYMXVM,VXX,2)
      Z=Z+ZINCR
      IF (Z.GE.DEGS2) GO TO 70
      GO TO 65
   70 Z=0.0
   75 CALL LABEL(XMNVM,YMNVM+Z,VXX,3)
      IF (LINYES) CALL MOVEA(XMNVM,TMERCA(YMNVM+Z))
      IF (LINYES) CALL DRAWA(XMXVM,TMERCA(YMNVM+Z))
      CALL LABEL(XMXVM,YMNVM+Z,VXX,4)
      Z=Z+ZINCR
      IF (Z.GE.DEGS1) GO TO 80
      GO TO 75
   80 IF (LINYES) GO TO 125
      LINYES=.TRUE.
C...ZINCR=999., THEREFORE BOUNDARY DRAWN ONLY
   85 CALL MOVEA (XMNVM,TYMNVM)
      CALL DRAWA (XMNVM,TYMXVM)
      CALL DRAWA (XMXVM,TYMXVM)
      CALL DRAWA (XMXVM,TYMNVM)
      CALL DRAWA (XMNVM,TYMNVM)
      IF (LINYES) GO TO 125
```

```
      CALL LABEL (XMNVM,YMNVM,VXX,3)
      CALL LABEL (XMNVM,YMXVM,VXX,3)
      CALL LABEL (XMNVM,TYMXVM,VXX,2)
      CALL LABEL (XMXVM,TYMXVM,VXX,2)
      CALL LABEL (XMXVM,YMXVM,VXX,4)
      CALL LABEL (XMXVM,YMNVM,VXX,4)
      CALL LABEL (XMXVM,TYMNVM,VXX,1)
      CALL LABEL (XMNVM,TYMNVM,VXX,1)
      GO TO 125
C...ZINCR .LT. 1, THEREFORE FRACTIONAL DEGREES TIC-ED
   90 TEMP1=XMNVM+Z
      TEMP2=FLOAT(INT(TEMP1))
      TENTH=.FALSE.
      IF (TEMP1-TEMP2.NE.0.) TENTH=.TRUE.
      IF ((Z.EQ.0.).OR.(Z.EQ.DEGS2)) GO TO 93
      IF (.NOT.TENTH) GO TO 93
      GO TO 97
   93 IF (TENTH) GO TO 95
      CALL LABEL (TEMP1,TYMNVM,VXX,1)
   95 CALL MOVEA (TEMP1,TYMNVM)
      CALL DRAWA (TEMP1,TYMXVM)
      IF (TENTH) GO TO 100
      CALL LABEL (TEMP1,TYMXVM,VXX,2)
      GO TO 100
   97 CALL MOVEA (TEMP1,TYMNVM)
      CALL DRAWA (TEMP1,TYMNVM+VXX)
      CALL MOVEA (TEMP1,TYMXVM-VXX)
      CALL DRAWA (TEMP1,TYMXVM)
  100 Z=Z+ZINCR
      Z=FLOAT(INT((Z+.01)*10.))/10.
      IF (Z.GT.DEGS2) GO TO 105
      GO TO 90
  105 Z=0.0
  107 TEMP1=YMNVM+Z
      TTEMP=TMERCA(TEMP1)
      TEMP2=FLOAT(INT(TEMP1))
      TENTH=.FALSE.
      IF (TEMP1-TEMP2.NE.0.) TENTH=.TRUE.
      IF ((Z.EQ.0.).OR.(Z.EQ.DEGS1)) GO TO 110
      IF (.NOT.TENTH) GO TO 110
      GO TO 115
  110 IF (TENTH) GO TO 113
      CALL LABEL (XMNVM,TEMP1,VXX,3)
  113 CALL MOVEA (XMNVM,TTEMP)
      CALL DRAWA (XMXVM,TTEMP)
      IF (TENTH) GO TO 120
      CALL LABEL (XMXVM,TEMP1,VXX,4)
      GO TO 120
  115 CALL MOVEA (XMNVM,TTEMP)
      CALL DRAWA (XMNVM+VXX,TTEMP)
      CALL MOVEA (XMXVM-VXX,TTEMP)
      CALL DRAWA (XMXVM,TTEMP)
  120 Z=Z+ZINCR
      Z=FLOAT(INT((Z+.01)*10.))/10.
      IF (Z.GT.DEGS1) GO TO 125
```

```
      GO TO 107
C...SET VIRTUAL AND SCREEN WINDOW FOR MAP
  125 CALL DWINDO (XMNVM,XMXVM,TYMNVM,TYMXVM)
      CALL TWINDO (MNXSM,MXXSM,MNYSM,MXYSM)
C...CALL STREAM IF MAPP IS GOING TO DRAW LAND MASS DATA BASE
      IF (LAND.NE.1) RETURN
      CALL STREAM
      RETURN
      END
C***********************************************************************
      SUBROUTINE STREAM
C...STREAM ACCESSES A SPECIFIED LAND MASS DATA BASE
C...(LOGICAL UNIT INPUT=01) IN THE STANDARD STREAM
C...FORMAT.  THEN READS AND DRAWS THE DATA BASE.
      DIMENSION DATA (2000)
C...DATA IS DIMENSIONED ARBITRARILY LARGE.
C...IT CAN BE CHANGED AS NECESSARY
      COMMON /VMAP/ XMNVM,XMXVM,YMNVM,YMXVM,TYMNVM,TYMXVM
      COMMON /SMAP/ MNXSM,MXXSM,MNYSM,MXYSM
C...SET SCREEN AND VIRTUAL WINDOWS FOR MAP,
C...IN CASE THEY HAVE BEEN CHANGED SINCE SUB MAPP WAS
C...CALLED.
      CALL DWINDO (XMNVM,XMXVM,TYMNVM,TYMXVM)
      CALL TWINDO (MNXSM,MXXSM,MNYSM,MXYSM)
C...READ AND DRAW FILE IN STREAMS
   10 READ (1,20,END=30) NPTS,(DATA(K),K=1,NPTS)
      DO 15 I=3,NPTS-1,2
      DATA(I)=DATA(I-2)+((DATA(I)-50.)/100.)
      DATA(I+1)=DATA(I-1)+((DATA(I+1)-50.)/100.)
   15 CONTINUE
      DO 17 I=1,NPTS-3,2
      IF (I.EQ.1) CALL MOVEA (DATA(I+1),TMERCA(DATA(I)))
      CALL DRAWA (DATA(I+3),TMERCA(DATA(I+2)))
   17 CONTINUE
      GO TO 10
C...FORMAT SPECIFICATION SET FOR 21 RECORDS OF DATA
C...MAXIMUM PER STREAM.  CHANGE AS NECESSARY.
   20 FORMAT (I5,2F7.2,1X,30F2.0/,20(40F2.0/))
   30 RETURN
      END
C***********************************************************************
      SUBROUTINE LABEL(X,Y,V,IFLG)
C...LABEL PLACES LAT-LON LINE LABEL OUTSIDE OF MAP
      DIMENSION LAB(3),IADE(10)
      DATA IADE/48,49,50,51,52,53,54,55,56,57/
      N=0
      IF (IFLG-3)1,2,2
C...LABEL FOR EAST OR WEST BOUNDARY
    1 I=IABS(INT(X))
      GO TO 3
C...LABEL FOR SOUTH OR NORTH BOUNDARY
    2 I=IABS(INT(Y))
    3 IF (I-10)4,5,5
C...LABEL .LT. 10
    4 LAB(1)=I
```

```
      N=1
      GO TO 8
    5 IF (I-100)6,7,7
C...LABEL .GE. 10 .AND. LABEL .LE. 99
    6 LAB(1)=INT(I/10)
      LAB(2)=MOD(I,10)
      N=2
      GO TO 8
C...LABEL .GE. 100
    7 LAB(1)=INT(I/100)
      LAB(2)=INT((I-100)/10)
      LAB(3)=MOD(I,10)
      N=3
C...DETERMINE EXACT LABEL LOCATION
    8 IF (IFLG-2)10,15,20
C...LABEL FOR WEST BOUNDARY
   10 CALL MOVEA(X,Y-V)
      CALL SEELOC(IX,IY)
      CALL MOVABS(IX-6*N/2,IY-10)
      GO TO 30
C...LABEL FOR EAST BOUNDARY
   15 CALL MOVEA(X,Y+V)
      CALL SEELOC(IX,IY)
      CALL MOVABS(IX-6*N/2,IY)
      GO TO 30
   20 IF (IFLG.EQ.4) GO TO 25
C...LABEL FOR SOUTH BOUNDARY
      CALL MOVEA(X-V,TMERCA(Y))
      CALL SEELOC(IX,IY)
      CALL MOVABS(IX-8*N,IY-4)
      GO TO 30
C...LABEL FOR NORTH BOUNDARY
   25 CALL MOVEA(X+V,TMERCA(Y))
      CALL SEELOC(IX,IY)
      CALL MOVABS(IX,IY-4)
C..."BEAM" IN POSITION, NOW OUTPUT LABEL
   30 DO 40 J=1,N
      K=LAB(J)
      CALL ANCHO(IADE(K+1))
   40 CONTINUE
      RETURN
      END
C**************************************************************************
      FUNCTION TMERCA(A)
C...TMERCA COMPUTES NEW VALUE FOR SOUTH OR NORTH
C...IF MERCATOR PROJECTION HAS BEEN SELECTED.
C...NOTE:  ADAPTED WITH PERMISSION FROM NODC GASHCONT SYSTEM
C...MERCTR SUBROUTINE.
      COMMON /PRJECT/ IPROJ
      IF (IPROJ.EQ.1) GO TO 10
      B=A*(3.141592654/180.0)
      SMAJOR=6378206.0
      SMINOR=6356583.0
      SDIFF=SMAJOR-SMINOR
      E=SQRT(SDIFF*SDIFF+SDIFF*2.0*SMINOR)/SMAJOR
```

```
      EHALF=E/2.0
      SINE=SIN(B)
      ESIN=E*SINE
      B=(ALOG((SINE+1.0)/COS(B)*((1.0-ESIN)/(1.0+ESIN))**EHALF))
      TMERCA=B*180.0/3.141592654
      RETURN
   10 TMERCA=A
      RETURN
      END
```

Appendix E

Gulf of Mexico Land Mass Data Base Listing

```
  246  29.68 -94.00 504749465047484748464947484748465047494748465047484749464847
5047484649474847484649474647494848485249535150525252515354525151525752525151 4954
5053515252515047504852495148494748454845494151495248505248515252525051525451 5552
6050484850495350494548464849494848484749484552505148524848494948484848504952 4852
4848474847524857485349474851485250524450464849524848504852474648494948484848 4949
4848484749485048535048474248524948484852494848484850494946485252525151525253 5250
50545150505149494036484948 48
   46  29.33 -94.73 504649494848504848495045494848474746464747474846494746485252
5760586753535052525152525152
  506  28.93 -95.30 454349495148443851525253525250485049484848484949505141345253
5250504846425250432850544950484647505154545651575253525451534543474443334943 4644
4541525052525560515252585052555548415249474146395250525151505049545350485354 5250
5048485050474846494748504847443553505248504851525243525053525047535049434848 4854
4950485347484652505347554852426048504740433749465149524855505047524848485049 5345
5048495045574550484848504752485044414849474651504949504851455248535050494435 4646
4950485750535352505352525350475348494950465150494746475051493440514550515250 5049
5250514649444848525251504938504751484948485250524850495150555155494938504865 4043
3443414749485045484948485048524750525253515050525750484844434647504852495248 5148
5547524847504554485347505052485050534855505352525255505345493648524850494848 4952
5051465247504748514752474848455047525048435047554852455144524352475143494748 4652
4950505150525052505146504949504848504852525152524852484844505052505148494850 4951
48524550485250484750505248515057475048484950505548525348545050524350
   18  28.40 -96.37 48505557545551535250484747454543
  250  28.35 -96.38 504855505048484947504846494748475048504848494948484748485048
4949504848504848504950454948504846505052494850495146494948484848474948484948 4549
4648494850494850474850484849484850484952434748484948464947484848454947484748 4850
5049454647504649494845504548485047492848455045523851475245524351475245524751 4352
4552435145524052405160505949564857485749584660476047654660495850625157525552 5551
5352585355545553575357545553535353525252535154525353535454535353525453535355 5252
525353535560525453535053525451535252
  560  29.68 -94.00 496350525350534850495448504851495248535054525656545251525253
4852495145504549464849485049504848454850495245534852535652575055526550585057 5157
5051505449604853476048584559475848584859495848584859505150525255525551535059 5051
5052525050485250505250525051514950485047494851485249525051515252504850455347 5250
5052505150525252485048484950505253555358505252515249515149525052505151525052 5053
5052455050485250504847494850505148504949485150495255535850524851505752534750 4549
4648505250564952455246504948515552574850495150555052485048524951515447514852 5053
4950464950534950485350524852455049514849504850485550535049475347535049464849 4850
4948504751484257485349585054465350525053495550525151495248534257505250535150 5455
5153475252535154575050535252485355475350505350525053495348524550485352545350 5451
5150495448504851495446534750435252535053575956585455534853505449514852505353 5250
5049525148544750535148525052525153525053495046474545495048525251505445485145 4948
4850484752605067525351524750484855534750455247605150524850525052475046534753 5052
4350505552535052485049475048485248504948484848495046475045495555575553535252 5051
5152524852505052
   64  29.76 -93.33 525052525048504851505452485352545051555253485349574853484949
434847484852485049484849504850474848495050525051485252525051505 2
   56  29.62 -92.00 464850494951485245625051485250535055525053525455515050485249
4848524850475247515050465247504747505048514749 48
  630  29.11 -89.25 425752525151525250525350505149524852555352504842535150525250
```

50524850505152504852505252505348504849475048514755505247534649455047535052485047
48475248535050535249504550414849535052485352494651495550505557505041535050495250
50515254465854475150525152495150505150524453505050525350525253505248535550525053
47535152544848495246515252455260505251505051525050425558505252505553514955555046
52505247514845445048494850455047534548484952484847455047435048484750484950455246
53475250525250484847504552485148555054544856485252525248515555525453505351545251
48525250505748505553505254505355525351525448484752505151505250534954484647505880
50655350504552504857495048534854495350535154505352534754485548515255525248535550
52545055465050515250485950534548525548525953615052525350525257515152545248515354
45514952505243504851495043474752455145574558454450455045514849474643526250585058
52595158525852595158505252505255535053565554555052485550475548504853445050525052
60535353424945584643494248454848504549525368536949504631494248405043504547545576
53675267485850584867408342674558485254525146545053495246484750485550506048524752
53534750454547574658525547504853455053475250524550475150524550484850475547554358
4755485447554553455340504844534855485749555147474552365652545056525951585058
34 29.90 -88.81 414847484849474855525351575255505748534952485052465149524352
4750
16 24.71 -97.65 5750535257535854374638475048
24 24.46 -97.68 54485149525052515552555253515052424843494348
18 30.25 -89.06 50504648504850495246505450535250
14 30.20 -88.98 505250515152504850494948
24 30.21 -88.56 50485247504552454845525351524955505548534852
18 30.20 -88.40 50495046504951465054505150544951
28 30.23 -88.31 5053505352575055485352525150494851475045494350474847
22 29.68 -85.21 4753485350525053535252525043484852495045
512 29.71 -85.00 505252505052524850525051535450535250505343454852545556635252
50525251566250535254505348574950504548535253535252485147504750485450485248565252
48485549555150525252485150545450505348505252535147505052505248484850505352525255
51574951485050524755485047584852485352524753515045574555495248514852455050524751
48554952465737514850495250534850485250524751505450534852475147525052465049514654
50534950485248484952485050514855525248585060495046504952504748505053485049534550
50524650475348545046495450514852504748505046455050524950485050484850505249505053
48505049485050514950505246474751475048504750485050494851454950484952485048484948
46504747475048505050474846494948484848504752455047504548465045494746475248524851
49545150505147504750505248554850495250515152545053525251535055495248514552475248
51535252525053504852495051515052495048535052485349504852455047484659525053485250
53505252505147495051555446504751475048504750454550474650505247434347495150494846
47505152525250514545485050575252505548505045495050415149524650494848495250514650
26 29.61 -85.08 495350525157545553585255515054574545413347385347
14 29.78 -84.66 525551535052494648495045
346 27.46 -82.65 475248554752455147504849485135594451455546524555445346554954
50535152475050515550584754505048515247525051525451504951505451564950504748475046
47504854375045484650455345505049475350535254485157545350535354555052485045434748
48504849474648475248484947505055485147545051525250525051485045524750504845524352
37514552415349524650445550525348525550534652495345605055485250534552485050524951
48524852495050514850485245525048495250534647495148505054485049504651475050544850
49534860485249514852485249514850484949465047515252505246535047475147524853485250
52495046485048504249485047514850495448514852525350605255515050525252485350525250
50514750515249525150495150555150505552525047525548515252515550524951
32 26.68 -82.18 525250554850475148505052485247505048435249514847654552485347
32 26.45 -82.01 464550485443514950485250525246535053495248505252515049535052
48 24.71 -82.88 526048504952465045465044495048484848494748505048524951485448
51525052525152495153525052505155
362 25.20 -80.50 515250525051475252505048514952485052525148524552505348504948
50495048504851504949485048534748504948484948464950484948464950485048494946485048

4949484850484847494848495048524848494848494850494843484850494948504848495048504848495048504849495048504850494845504848485249474850485049504853485349524852484650
4752485249505151475050525252484850485350494951505048524852495546465047544850505149505052515049505048515050494950515150495048524852505049514849504852485050514950
5152494851505049524852485049465050515052495051524948515050485049504852485049485050514952485050485249504848485250504948484848494948485048524946505248484850495048
50484949505150525052505151525250485249505151505250525251525050504852505250515252
50525151505250525550
172 25.50 -80.33 5352535154495151525252525350535154555152525053505452515052515052484849504852485049484850485049525150525048525550525055505350595058515850595258525850595158525850595058505847594858475746514752505547604558455948584755485052
4551425441534253455247534154594858455847594658475848594758455845594558475848594858495853525252535150544853495350544856 45
274 26.06 -97.16 4451405045494848504857524847484847505047504543554951555451515254434839474146354237423341454947465750605555545855504852475153545053474850524353484745505248534254534652495048405248485448504753475343504845505254475046484950
5048534747504658475048524950504755484648475047475157455149494648495046485054495148544850494341484049475052514852525241484748454943484550425345524547455042464549
45504151455264526051615255526550555144524648354940483948334942513649404845484249355142523852355045484347524843484549474847484852485260533547455045514452465245514552
410 22.21 -97.78 4550405335593958405843594358455045484247434747484250465244534053465452485348524947504851485250525051445545524852485047514352495046523960435650494851505250524455405841584555455439603661425547524150395241554748485045534749
43584955465250515252435049534550465450584752435042514365495546505540494755475146
4550505447554853495250535052485157524553485448584958516749654558505448504751465249574866405448514552475745534552475148524758505953665775535953585255525335159576652595053507248665567576251505451465247524751515055495448485250515767465255505063
5467516650544455465350534750484752455047475545584375525553525251556250575853575355505345484752455348474450455755555553555053576361706467585858546953535052485350555262675853535254485152644653506352555254505548605065535150595451515550595455515559535657605157525355595258536752 66
82 24.95 -80.61 5156545251535252535254535052535353535049545153545351545253535352545253515552454747485048464947484747484548534748484949485047515054485353535254555352534854505550